
ABSTRACT

This paper describes a graph-drawing package that has been implemented as an extension to the
MetaPost graphics language. MetaPost has a powerful macro facility for implementing such exten-
sions. There are also some new language features that support the graph macros. Existing features
for generating and manipulating pictures allow the user to do things that would be di�cult to achieve
in a stand-alone graph package.

1 Introduction

MetaPost is a batch-oriented graphics language based on Knuth's METAFONT1, but with PostScript2

output and numerous features for integrating text and graphics. The author has tried to make this
paper as independent as possible of the user's manual [5], but fully appreciating all the material
requires some knowledge of the MetaPost language.

We concentrate on the mechanics of producing particular kinds of graphs because the question
of what type of graph is best in a given situation is covered elsewhere; e.g., Cleveland [2, 4, 3] and
Tufte [11]. The goal is to provide at least the power of UNIX3 grap [1], but within the MetaPost
language. Hence the package is implemented using MetaPost's powerful macro facility.

The graph macros provide the following functionality:

1. Automatic scaling

2. Automatic generation and labeling of tick marks or grid lines

3. Multiple coordinate systems

4. Linear and logarithmic scales

5. Separate data �les

6. Ability to handle numbers outside the usual range

7. Arbitrary plotting symbols

8. Drawing, �lling, and labeling commands for graphs

In addition to these items, the user also has access to all the features described in the MetaPost
user's manual [5]. These include access to almost all the features of PostScriptR, ability to use and
manipulate typeset text, ability to solve linear equations, and data types for points, curves, pictures,
and coordinate transformations.

Section 2 describes the graph macros from a user's perspective and presents several examples.
Sections 3 and 4 discuss auxiliary packages for manipulating and typesetting numbers and Section 5
gives some concluding remarks. Appendix A summarizes the graph-drawing macros, and Appendix B
describes some recent additions to the MetaPost language that have not been presented elsewhere.

2 Using the Graph Macros

A MetaPost input �le that uses the graph macros should begin with

input graph

This reads a macro �le graph.mp and de�nes the graph-drawing commands explained below. The
rest of the �le should be one or more instances of

beginfig(h�gure numberi);
hgraphics commandsi endfig;

1
METAFONT is a trademark of Addison Wesley Publishing Company.

2PostScript is a registered trademark of Adobe Systems Inc.
3UNIX is a registered trademark of UNIX System Laboratories, Inc.

followed by end.

The following hgraphics commandsi su�ce to generate the graph in Figure 1 from the data �le
agepop91.d:

draw begingraph(3in,2in);

gdraw "agepop91.d";

endgraph;

(Each line of agepop91.d gives an age followed the estimated number of Americans of that age in
1991 [10].)

Figure 1: A graph of the 1991 age distribution in the United States

2.1 Basic Graph-Drawing Commands

All graphs should begin with

begingraph(hwidthi,hheighti);

and end with endgraph. This is syntactically a hpicture expressioni, so it should be preceded by
draw and followed by a semicolon as in the example.4 The hwidthi and hheighti give the dimensions
of the graph itself without the axis labels.

The command
gdraw hexpressioni hoption listi

draws a graph line. If the hexpressioni is of type string, it names a data �le; otherwise it is a path
that gives the function to draw. The hoption listi is zero or more drawing options

withpenhpen expressioni j withcolorhcolor expressioni j dashedhpicture expressioni

that give the line width, color, or dash pattern as explained in the User's Manual [5].

In addition to the standard drawing options, the hoption listi in a gdraw statement can contain

plot hpicture expressioni

4See the User's Manual [5] for explanations of draw commands and syntactic elements like hpicture expressioni.

The hpicture expressioni gives a plotting symbol to be drawn at each path knot. The plot option
suppresses line drawing so that5

gdraw "agepop91.d" plot btex $\bullet$ etex

generates only bullets as shown in Figure 2. (Following the plot option with a withpen option
would cause the line to reappear superimposed on the plotting symbols.)

Figure 2: The 1991 age distribution plotted with bullets

The glabel and gdotlabel commands add labels to a graph. The syntax for glabel is

glabel: hlabel su�xi(hstring or picture expressioni, hlocationi) hoption listi

where hlocationi identi�es the location being labeled and hlabel su�xi tells how the label is o�set
relative to that location. The gdotlabel command is identical, except it marks the location with
a dot. A hlabel su�xi is as in plain MetaPost: hemptyi centers the label on the location; lft, rt,
top, bot o�set the label horizontally or vertically; and ulft, urt, llft, lrt give diagonal o�sets.
The hlocationi can be a pair of graph coordinates, a knot number on the last gdraw path, or the
special location OUT. Thus

gdotlabel.top(btex $(50,0)$ etex, 50,0)

would put a dot at graph coordinates (50,0) and place the typeset text \(50; 0)" above it. Alter-
natively,

glabel.ulft("Knot3", 3)

typesets the string "Knot3" and places it above and to the left of Knot 3 of the last gdraw path.
(The knot number 3 the path's \time" parameter [5, Section 8.2].)

The hlocationi OUT places a label relative to the whole graph. For example, replacing \gdraw
"agepop91.d"" with

glabel.lft(btex \vbox{\hbox{Population} \hbox{in millions}} etex, OUT);

glabel.bot(btex Age in years etex, OUT);

gdraw "agepopm.d";

in the input for Figure 1 generates Figure 3. This improves the graph by adding axis labels and
using a new data �le agepopm.d where the populations have been divided by one million to avoid

Figure 3: An improved version of the 1991 age distribution graph

large numbers. We shall see later that simple transformations such as this can be achieved without
generating new data �les.

All avors of TEX can handle multi-line labels via the \hbox within \vbox arrangement used
above, but LaTEX users will �nd it more natural to use the tabular environment [9]. Tro� user's
can use no�ll mode:

btex .nf

Population

in millions etex

2.2 Coordinate Systems

The graph macros automatically shift and rescale coordinates from data �les, gdraw paths, and
glabel locations to �t the graph. Whether the range of y coordinates is 0.64 to 4.6 or 640,000 to
4,600,000, they get scaled to �ll about 88% of the height speci�ed in the begingraph statement. Of
course line widths, labels, and plotting symbols are not rescaled.

The setrange command controls the shifting and rescaling process by specifying the minimum
and maximum graph coordinates:

setrange(hcoordinatesi, hcoordinatesi)

where

hcoordinatesi ! hpair expressioni
j hnumeric or string expressioni,hnumeric or string expressioni

The �rst hcoordinatesi give (xmin; ymin) and the second give (xmax; ymax). The lines x = xmin,
x = xmax, y = ymin , and y = ymax de�ne the rectangular frame around the graph in Figures 1{3.
For example, an adding a statement

setrange(origin, whatever,whatever)

to the input for Figure 3 yields Figure 4. The �rst hcoordinatesi are given by the prede�ned pair
constant origin, and the other coordinates are left unspeci�ed. Any unknown value would work as
well, but whatever is the standard MetaPost representation for an anonymous unknown value.

5Tro� users should replace btex $\bullet$ etex with btex \(bu etex.

draw begingraph(3in,2in);

glabel.lft(btex \vbox{\hbox{Population} \hbox{in millions}} etex, OUT);

glabel.bot(btex Age in years etex, OUT);

setrange(origin, whatever,whatever);

gdraw "agepopm.d";

endgraph;

Figure 4: The 1991 age distribution graph and the input that creates it.

Notice that the syntax for setrange allows coordinate values to be given as strings. Many
commands in the graph package allow this option. It is provided because the MetaPost language
uses �xed point numbers that must be less than 32768. This limitation is not as serious as it sounds
because good graph design dictates that coordinate values should be \of reasonable magnitude" [2,
11]. If you really want x and y to range from 0 to 1,000,000,

setrange(origin, "1e6","1e6")

does the job. Any �xed or oating point representation is acceptable as long as the exponent is
introduced by the letter \e".

Coordinate systems need not be linear. The setcoords command allows either or both axes to
have logarithmic spacing:

hcoordinate settingi ! setcoords(hcoordinate typei, hcoordinate typei)
hcoordinate typei ! log j linear j -log j -linear

A negative hcoordinate typei makes x (or y) run backwards so it is largest on the left side (or bottom)
of the graph.

Figure 5 graphs execution times for two matrix multiplication algorithms using

setcoords(log,log)

to specify logarithmic spacing on both axes. The data �le matmul.d gives timings for both algo-
rithms:

20 .007861 standard MM: size, seconds

30 .022051

40 .050391

60 .15922

80 .4031

120 1.53

160 3.915

240 18.55

320 78.28

480 279.24

20 .006611 Strassen: size, seconds

30 .020820

40 .049219

60 .163281

80 .3975

120 1.3125

160 3.04

240 9.95

320 22.17

480 72.60
A blank line in a data �le ends a data set. Subsequent gdraw commands access additional data sets
by just naming the same data �le again. Since each line gives one x coordinate and one y coordinate,
commentary material after the second data �eld on a line is ignored.

Standard

Strassen

draw begingraph(2.3in,2in);

setcoords(log,log);

glabel.lft(btex Seconds etex,OUT);

glabel.bot(btex Matrix size etex,

OUT);

gdraw "matmul.d" dashed evenly;

glabel.ulft(btex Standard etex,8);

gdraw "matmul.d";

glabel.lrt(btex Strassen etex,7);

endgraph;

Figure 5: Timings for two matrix multiplication algorithms with the corresponding MetaPost input.

Placing a setcoords command between two gdraw commands graphs two functions in di�erent
coordinate systems as shown in Figure 6. Whenever you give a setcoords command, the interpreter
examines what has been drawn, selects appropriate x and y ranges, and scales everything to �t.
Everything drawn afterward is in a new coordinate system that need not have anything in common
with the old coordinates unless setrange commands enforce similar coordinate ranges. For instance,
the two setrange commands force both coordinate systems to have x ranging from 80 to 90 and
y starting at 0.

When you use multiple coordinate systems, you have to specify where the axis labels go. The
default is to put tick marks on the bottom and the left side of the frame using the coordinate system
in e�ect when the endgraph command is interpreted. Figure 6 uses the

autogrid(,otick.lft)

to label the left side of the graph with the y coordinates in e�ect before the setcoords command.



Drawing Graphs with MetaPost 7

draw begingraph(6.5cm,4.5cm);

setrange(80,0, 90,whatever);

glabel.bot(btex Year etex, OUT);

glabel.lft(btex \vbox{\hbox{Emissions in} \hbox{thousands of}

\hbox{metric tons} \hbox{(heavy line)}}etex, OUT);

gdraw "lead.d" withpen pencircle scaled 1.5pt;

autogrid(,otick.lft);

setcoords(linear,linear);

setrange(80,0, 90,whatever);

glabel.rt(btex \vbox{\hbox{Micrograms} \hbox{per cubic}

\hbox{meter of air} \hbox{(thin line)}}etex, OUT);

gdraw "lead.d";

autogrid(otick.bot,otick.rt);

endgraph;

Figure 6: Annual lead emissions and average level at atmospheric monitoring stations in the United
States. The MetaPost input is shown above the graph.

This suppresses the default axis labels, so another autogrid command is needed to label the bottom
and right sides of the graph using the new coordinate system. The general syntax is

autogrid(haxis label commandi, haxis label commandi) hoption listi

where

haxis label commandi ! hemptyi j hgrid or ticki hlabel su�xi
hgrid or ticki ! grid j itick j otick

The hlabel su�xi should be lft, rt, top, or bot.

The �rst argument to autogrid tells how to label the x axis and the second argument does the
same for y. An hemptyi argument suppresses labeling for that axis. Otherwise, the hlabel su�xi
tells which side of the graph gets the numeric label. Be careful to use bot or top for the x axis and
lft or rt for the y axis. Use otick for outward tick marks, itick for inward tick marks, and grid

for grid lines. The hoption listi tells how to draw the tick marks or grid lines. Grid lines tend to
be a little overpowering, so it is a good idea to give a withcolor option to make them light gray so
they do not make the graph too busy.

2.3 Explicit Grids and Framing

In case autogrid is not exible enough, axis label commands generate grid lines or tick marks one
at a time. The syntax is

hgrid or ticki.hlabel su�xi(hlabel formati, hnumeric or string expressioni) hoption listi

where hgrid or ticki and hlabel su�xi are as in autogrid, and hlabel formati is either a format string
like "%g" or a picture containing the typeset numeric label.

The axis label commands use a macro

format(hformat stringi, hnumeric or string expressioni)

to typeset numeric labels. Full details appear in Section 4, but when the hformat stringi is "%g",
it uses decimal notation unless the number is large enough or small enough to require scienti�c
notation.

The example in Figure 7 invokes
format("%g",y)

explicitly so that grid lines can be placed at transformed coordinates. It de�nes the transformation
newy(y) = y=75 + ln y and shows that this function increases almost linearly.6 This is a little like
using logarithmic y-coordinates, except that y is mapped to y=75 + lny instead of just ln y.

Figure 7 uses the command

frame.hlabel su�xi hoption listi

to draw a special frame around the graph. In this case the hlabel su�xi is llft to draw just the
bottom and left sides of the frame. Su�xes lrt, ulft, and urt draw other combinations of two
sides; su�xes lft, rt, top, bot draw one side, and hemptyi draws the whole frame. For example

frame dashed evenly

6The manual [5] explains how vardef de�nes functions and mlog computes logarithms.

vardef newy(expr y) = (256/75)*y + mlog y enddef;

draw begingraph(3in,2in);

glabel.lft(btex \vbox{\hbox{Population} \hbox{in millions}} etex, OUT);

gdraw "ttimepop.d";

for y=5,10,20,50,100,150,200,250:

grid.lft(format("%g",y), newy(y)) withcolor .85white;

endfor

autogrid(grid.bot,) withcolor .85white;

frame.llft;

endgraph;

Figure 7: Population of the United States in millions versus time with the population re-expressed
as p=75 + lnp. The MetaPost input shown above the graph assumes a data �le ttimepop.d that
gives (year, p=75 + ln p) pairs.

draws all four sides with dashed lines. The default four-sided frame is drawn only when there is no
explicit frame command.

To label an axis as autogrid does but with the labels transformed somehow, use

auto.x or auto.y

for positioning tick marks or grid lines. These macros produce comma-separated lists for use in for

loops. Any x or y values in these lists that cannot be represented accurately within MetaPost's
�xed-point number system are given as strings. A standard macro package that is loaded via

input sarith

de�nes arithmetic operators that work on numbers or strings. Binary operators Sadd, Ssub, Smul,
and Sdiv do addition, subtraction multiplication, and division.

One possible application is rescaling data. Figure 4 used a special data �le agepopm.d that had
y values divided by one million. This could be avoided by replacing \gdraw "agepopm.d"" by

gdraw "agepop91.d";

for u=auto.y: otick.lft(format("%g",u Sdiv "1e6"), u); endfor

autogrid(otick.bot,)

2.4 Processing Data Files

The most general tool for processing data �les is the gdata command:

gdata(hstring expressioni, hvariablei, hcommandsi)

It takes a �le name, a variable v, and a list of commands to be executed for each line of the data
�le. The commands are executed with i set to the input line number and strings v1, v2, v3, : : : set
to the input �elds on the current line. A null string marks the end of the v array.

Using a glabel command inside of gdata generates a scatter plot as shown in Figure 8. The
data �le countries.d begins

20.910 75.7 US

1.831 66.7 Alg

where the last �eld in each line gives the label to be plotted. Setting defaultfont in the �rst line
of input selects a small font for these labels. Without these labels, no gdata command would be
needed. Replacing the gdata command with

gdraw "countries.d" plot btex$\circ$etex

would change the abbreviated country names to open circles.

Both gdraw and gdata ignore an optional initial `%' on each input line, parse data �elds separated
by white space, and stop if they encounter an input line with no data �elds. Leading percent signs
make graph data look like MetaPost comments so that numeric data can be placed at the beginning
of a MetaPost input �le.

It is often useful to construct one or more paths when reading a data �le with gdata. The
augment command is designed for this:

augment.hpath variablei(hcoordinatesi)

If the path variable does not have a known value, it becomes a path of length zero at the given
coordinates; otherwise a line segment to the given coordinates is appended to the path. The hcoor-
dinatesi may be a pair expression or any combination of strings and numerics as explained at the
beginning of Section 2.2.

defaultfont:="cmr7";

draw begingraph(3in,2in);

glabel.lft(btex \vbox{\hbox{Life}\hbox{expectancy}} etex, OUT);

glabel.bot(btex Per capita G.N.P. (thousands of dollars) etex, OUT);

setcoords(log,linear);

gdata("countries.d", s,

glabel(s3, s1, s2);

)

endgraph;

Figure 8: A scatter plot and the commands that generated it

If a �le timepop.d gives t, p pairs, augment can be used like this to graph newy(p) versus t:

path p;

gdata("timepop.d", s, augment.p(s1, newy(scantokens s2)); );

gdraw p;

(MetaPost's scantokens primitive interprets a string as if it were the contents of an input �le. This
�nds the numeric value of data �eld s2.)

Figure 9 shows how to use augment to read multiple column data and make multiple paths.
Paths p2, p3, p4, p5 give cumulative totals for columns 2 through 5 and pictures lab2 through lab5

give corresponding labels. The expression

image(unfill bbox lab[j]; draw lab[j])

executes the given drawing commands and returns the resulting picture: \unfill bbox lab[j]"
puts down a white background and \draw lab[j]" puts the label on the background. The gfill

command is just like gdraw, except it takes a cyclic path and �lls the interior with a solid color. The
color is black unless a withcolor clause speci�es another color. See the manual [5] for explanations
of for loops, arrays, colors, and path construction operators like --, cycle, and reverse.

3 Manipulating Big Numbers

MetaPost inherits a �xed-point number system fromKnuth's METAFONT [8]. Numbers are expressed
in multiples of 2�16 and they must have absolute value less than 32768. Knuth chose this system

draw begingraph(3in,2in);

glabel.lft(btex \vbox{\hbox{Quadrillions}\hbox{of BTU}} etex, OUT);

path p[];

numeric t;

gdata("energy.d", $,

t:=0; augment.p1($1,0);

for j=2 upto 5:

t:=t+scantokens $[j]; augment.p[j]($1,t);

endfor)

picture lab[];

lab2=btex coal etex; lab3=btex crude oil etex;

lab4=btex natural gas etex; lab5=btex hydroelectric etex;

for j=5 downto 2:

gfill p[j]--reverse p[j-1]--cycle withcolor .16j*white;

glabel.lft(image(unfill bbox lab[j]; draw lab[j]), .7+length p[j]);

endfor

endgraph;

hydroelectric

natural gas

crude oil

coal

1900 1920 1940 1960

20

40

60

Quadrillions
of BTU

Figure 9: A graph of U.S. annual energy production and the commands that generated it

because it is perfectly adequate for font design, and it guaranteed to give identical results on all types
of computers. Fixed-point numbers are seldom a problem in MetaPost because all computations are
based on coordinates that are limited by the size the paper on which the output is to be printed. This
does not hold for the input data in a graph-drawing application. Although graphs look best when
coordinate axes are labeled with numbers of reasonable magnitude, the strict limits of �xed-point
arithmetic would be inconvenient.

A simple way to handle large numbers is to include the line

input sarith

and then use binary operators Sadd, Ssub, Smul, and Sdiv in place of +, -, *, and /. These operators
are ine�cient but very exible. They accept numbers or strings and return strings in exponential
notation with the exponent marked by \e"; e.g., "6.7e-11" means 6:7� 10�11.

The unary operator7

Sabs hstringi

�nds a string the represents the absolute value. Binary operators Sleq and Sneq perform numeric
comparisons on strings and return boolean results.

The operation
Scvnum hstringi

�nds the numeric value for a string if this can be done without overowing MetaPost's �xed-point
number system. If the string does not contain \e", it is much more e�cient to use the primitive
operation

scantokens hstringi

The above operators are based on a low-level package that manipulates numbers in \Mlog form."
A number x in Mlog form represents

�
216x

; where � = �e2
�24

:

Any value between 1:61� 10�28 and 3:88� 1055 can be represented this way. (There is a constant
Mten such that k � Mten represents 10k for any integer k in the interval [�29; 55].)

The main reason for mentioning Mlog form is that it allows graph data to be manipulated as a
MetaPost path. The function

Mreadpath(h�le namei)

reads a data �le and returns a path where all the coordinates are in Mlog form. An internal variable
Gpaths determines whether gdraw and gfill expect paths to be given in Mlog form. For example,
this graphs the data in agepop91.d with y coordinates divided by one million:

interim Gpaths:=log;

gdraw Mreadpath("agepop91.d") shifted (0,-6*Mten);

4 Typesetting Numbers

The graph package needs to compute axis labels and then typeset them. The macro

format(hstring expressioni, hnumeric or string expressioni)

7The argument to a unary operator need not be parenthesized unless it is an expression involving binary operators.

does this. You must �rst input graph or input format to load the macro �le. The macro takes a
format string and a number to typeset and returns a picture containing the typeset result. Thus

format("%g",2+2) yields 4

and
format("%3g","6.022e23") yields 6.02�1023

A format string consists of

� an optional initial string not containing a percent sign,

� a percent sign,

� an optional numeric precision p,

� one of the conversion letters e, f, g, G,

� an optional �nal string �.

The initial and �nal strings are typeset in the default font (usually cmr10), and the typeset number
is placed between them. For the e and g formats, the precision p is the number of signi�cant digits
allowed after rounding; for f and G, the number is rounded to the nearest multiple of 10�p. If the
precision is not speci�ed, the default is p = 3. The e format always uses scienti�c notation and the
f format uses ordinary decimal notation but reverts to scienti�c notation if the number is at least
10000. The g and G formats also revert to scienti�c notation for non-zero numbers of magnitude less
than 0.001.

The format macro needs a set of templates to determine what font to use, how to position the
exponent, etc. The templates are normally initialized automatically, but it is possible to set them
explicitly by passing �ve picture expressions to init numbers. For instance, the default de�nition
for TEX users is

init_numbers(btex$-$etex, btex$1$etex, btex${\times}10$etex,

btex${}^-$etex, btex${}^2$etex)

The �rst argument tells how to typeset a leading minus sign; the second argument is an example of
a 1-digit mantissa; third comes whatever to put after the mantissa in scienti�c notation; next come
a leading minus sign for the exponent and a sample 1-digit exponent.

Picture variable Fe_plus gives a leading plus sign for positive numbers, and Fe_base gives
whatever should precede the exponent when typesetting a power of ten. Calling init_numbers

initializes Fe_plus to an empty picture and constructs Fe_base from its second and third arguments.

5 Conclusion

The graph package makes it convenient to generate graphs from within the MetaPost language. The
primary bene�ts are the power of the MetaPost language and its ability to interact with TEX or tro�
for typesetting labels. Typeset labels can be stored in picture variables and manipulated in various
ways such measuring the bounding box and providing a white background.

We have seen how to generate shaded regions and control line width, color, and styles of dashed
lines. Numerous other variations are possible. The full MetaPost language [5] provides many other
potentially useful features. It also has enough computing power to be useful for generating and
processing data.

A Summary of the Graph Package

In the following descriptions, italic letters such as w and h denote expression parameters and words
in angle brackets denote other syntactic elements. Unless speci�ed otherwise, expression parameters
can be either numerics or strings. An hoption listi is a list of drawing options such as withcolor
.5white or dashed evenly; a hlabel su�xi is one of lft, rt, top, bot, ulft, urt, llft, lrt.

A.1 Graph Administration

begingraph(w,h) Begin a new graph with the frame width and height given by numeric parameters
w and h.

endgraph End a graph and return the resulting picture.

setcoords(tx, ty) Set up a new coordinate system as speci�ed by numeric ags tx, ty. Flag values
are �linear and �log.

setrange(hcoordinatesi, hcoordinatesi) Set the lower and upper limits for the current coordinate
system. Each hcoordinatesi can be a single pair expression or two numeric or string expressions.

A.2 Drawing and Labeling

All of the drawing and labeling commands can be followed by an hoption listi. In addition to the
usual MetaPost drawing options, the list can contain a plot hpicturei clause to plot a speci�ed
picture at each data point.

The drawing and labeling commands are closely related to a set of similarly named commands
in plain MetaPost. The gdrawarrow and gdrawdblarrow commands are included to maintain this
relationship.

gdotlabel.hlabel su�xi(p, hlocationi) This is like glabel except it also puts a dot at the location
being labeled.

gdraw p Draw path p, or if p is a string, read coordinate pairs from �le p and draw a polygonal line
through them.

gdrawarrow p This is like dgraw p except it adds an arrowhead at the end of the path.

gdrawdblarrow p This is like dgraw p except it adds an arrowheads at each end of the path.

gfill p Fill cyclic path p or read coordinates from the �le named by string p and �ll the resulting
polygonal outline.

glabel.hlabel su�xi(p, hlocationi) If p is not a picture, it should be a string. Typeset it using
defaultfont, then place it near the given location and o�set as speci�ed by the hlabel su�xi.
The hlocationi can be x and y coordinates, a pair giving x and y, a numerc value giving a time
on the last path drawn, or OUT to label the outside of the graph.

A.3 Grids, Tick Marks, and Framing

auto.hx or yi Generate default x or y coordinates for tick marks.

autogrid(haxis label commandi, haxis label commandi) Draw default axis labels using the speci-
�ed commands for the x and y axes. An haxis label commandi may be hemptyi or it may be
itick, otick, or grid followed by a hlabel su�xi.

frame.hlabel su�xi hoption listi Draw a frame around the graph, or draw the part of the frame
speci�ed by the hlabel su�xi.

grid.hlabel su�xi(f ,z) Draw a grid line across the graph from the side speci�ed by the hlabel
su�xi, and label it there using format string f and coordinate value z. If f is a picture, it
gives the label.

itick.hlabel su�xi(f ,z) This is like grid except it draws an inward tick mark.

otick.hlabel su�xi(f ,z) This is like grid except it draws an outward tick mark.

A.4 Miscellaneous Commands

augment.hvariablei(hcoordinatesi) Append hcoordinatesi to the path stored in hvariablei.

format(f,x) Typeset x according to format string f and return the resulting picture.

gdata(f, hvariablei, hcommandsi) Read the �le named by string f and execute hcommandsi for
each input line using the hvariablei as an array to store data �elds.

init numbers(s,m,x, t, e) Provide �ve pictures as templates for future format operations: s is
a leading minus; m is a sample mantissa; x follows the mantissa; t is a leading minus for the
exponent e.

Mreadpath(f) Read a path for the data �le named by string f and return it in \Mlog form".

A.5 Arithmetic on Numeric Strings

It is necessary to input sarith before using the following macros:

Sabs x Compute jxj and return a numeric string.

x Sadd y Compute x+ y and return a numeric string.

Scvnum x Return the numeric value for string x.

x Sdiv y Compute x=y and return a numeric string.

x Sleq y Return the boolean result of the comparison x � y.

x Smul y Compute x � y and return a numeric string.

x Sneq y Return the boolean result of the comparison x 6= y.

x Ssub y Compute x� y and return a numeric string.

A.6 Internal Variables and Constants

Autoform Format string used by autogrid. Default: "%g".

Fe base What precedes the exponent when typesetting a power of ten.

Fe plus Picture of the leading plus sign for positive exponents.

Gmarks Minimum number of tick marks per axis for auto and autogrid. Default: 4.

Gminlog Minimum largest/smallest ratio for logarithmic spacing with auto and autogrid. Default:
3.0.

Gpaths Code for coordinates used in gdraw and gfill paths: linear for standard form, log for
\Mlog form".

Mten The \Mlog form" for 10.0

B New Language Features

The graph.mpmacros and the arithmetic routines in marith.mp and sarith.mp use various language
features that were introduced in Version 0.60 of the MetaPost language. We summarize these features
here because they are not covered in existing documentation [5, 6]. Also new is the built-in macro

image(hdrawing commandsi)

that was used in Section 2.4 to �nd the picture produced by a sequence of drawing commands.

B.1 Reading and Writing Files

A new operator
readfrom h�le namei

returns a string giving the next line of input from the named �le. The h�le namei can be any primary
expression of type string. If the �le has ended or cannot be read, the result is a string consisting
of a single null character. The preloaded plain macro package introduces the name EOF for this
string. After readfrom has returned EOF, additional reads from the same �le cause the �le to be
reread from the start.

The opposite of readfrom is the command

write hstring expressioni toh�le namei

This writes a line of text to the speci�ed output �le, opening the �le �rst if necessary. All such �les
are closed automatically when the program terminates. They can also be closed explicitly by using
EOF as the hstring expressioni. The only way to tell if a write command has succeeded is to close
the �le and use readfrom to look at it.

B.2 Extracting Information from Pictures

MetaPost pictures are composed of stroked lines, �lled outlines, pieces of typeset text, clipping paths,
and setbounds paths. (A setbounds path gives an arti�cial bounding box as is needed for TEX
output.) A picture can have many components of each type. They can be accessed via an iteration
of the form

for hsymbolic tokeni within hpicture expressioni: hloop texti endfor

The hloop texti can be anything that is balanced with respect to for and endfor. The hsymbolic
tokeni is a loop variable that scans the components of the picture in the order in which they were
drawn. The component for a clipping or setbounds path includes everything the path applies to.
Thus if a single clipping or setbounds path applies to everything in the hpicture expressioni, the
whole picture could be thought of as one big component. In order to make the contents of such a
picture accessible, the for: : :within iteration ignores the enclosing clipping or setbounds path in
this case.

Once the for: : :within iteration has found a picture component, there are numerous operators
for identifying it and extracting relevant information. The operator

stroked hprimary expressioni

tests whether the expression is a known picture whose �rst component is a stroked line. Similarly,
the filled and textual operators return true if the �rst component is a �lled outline or a piece
of typeset text. The clipped and bounded operators test whether the argument is a known picture
that starts with a clipping path or a setbounds path. This is true if the �rst component is clipped
or bounded or if the entire picture is enclosed in a clipping or setbounds path.

There are also numerous part extraction operators that test the �rst component of a picture. If
p is a picture and stroked p is true, pathpart p is the path describing the line that got stroked,
penpart p is the pen that was used, dashpart p is the dash pattern, and the color is

(redpart p, greenpart p, bluepart p)

If the line is not dashed, dashpart p returns an empty picture.

The same part extraction operators work when filled p is true, except that dashpart p is
not meaningful in that case. For text components, textual p is true, textpart p gives the text
that got typeset, fontpart p gives the font that was used, and xpart p, ypart p, xxpart p,
xypart p, yxpart p, yypart p tell how the text has been shifted, rotated, and scaled. The redpart,
greenpart, and bluepart operators also work for text components.

When clipped p or bounded p is true, pathpart p gives the clipping or setbounds path and
the other part extraction operators are not meaningful. Such non-meaningful part extractions do
not generate errors|they return null values instead: the trivial path (0,0) for pathpart, nullpen
for penpart, an empty picture for dashpart, zero for redpart, greenpart, bluepart, and the null
string for textpart or fontpart.

One �nal operator for extracting information from a picture is

length hpicture primaryi

This returns the number of components that a for: : :within iteration would �nd.

B.3 Other New Features

The marith.mp and sarith.mp packages use numbers of magnitude 4096 more. Since such numbers
can cause overow problems in MetaPost's linear equation solving and path �tting algorithms, they
are normally allowed only as intermediate results. This limitation is removed when the internal
variable warningcheck is zero. In earlier versions of MetaPost, the limitation could be removed for
variables but explicit constants were always restricted to be less than 4096.

For completeness, we also mention one other new feature of MetaPost Version 0.60. When
TEX material is included in a picture via the btex: : :etex feature, the thickness of horizontal and
vertical rules gets rounded to exactly the right number of pixels; i.e., interpreting MetaPost output
according to the PostScriptR scan conversion rules [7] makes the pixel width equal to the ceiling
of the unrounded width. In fact, a similar relationship holds for all line widths. The generated
PostScript sets line widths by �rst transforming to device coordinates and rounding appropriately.

Atts.
References

